Práctica 9 (Cálculo)

Integral doble

```
In[1]:= Clear["Global`*"]
```

■ Integral doble en dominios rectangulares.

Consideramos una función f:D⟶ℝ integrable Riemann, donde D=[a,b]×[c,d] es un dominio rectangular. La integral de f en el dominio D se calcula mediante el comando Integrate. En el siguiente ejemplo se integra la función f en el dominio [1,7]×[2,3], es decir x∈[1,7], y∈[2,3].

```
In[2]:= f[x_, y_] := 2 x + 7 y + 2
```

```
In[3]:= Plot3D[f[x, y], {x, 1, 7}, {y, 2, 3}]
```




```
Out[3]= - SurfaceGraphics -
```

```
In[4]:= Integrate[f[x, y], {x, 1, 7}, {y, 2, 3}]
```

```
Out[4]= 165
```

```
In[5]:= Integrate[f[x, y], {y, 2, 3}, {x, 1, 7}]

Out[5]= 165
```

Al efectuar las integrales sin emplear el ordenador, se usa el Teorema de Fubini y se efectua la integral en una de las variables, y luego se integra en la otra variable. Este proceso también se puede realizar con *Mathematica.*

Primero calculamos la integral de f entre x=1 y x=7. Esta integral depende de y, la notamos F[y]. Luego calculamos la integral de esta función entre y=2 e y=3.

```
In[6]:= F[y_] = Integrate[f[x, y], {x, 1, 7}]

Out[6]= 60 + 42 y
```

```
In[7]:= Integrate[F[y], {y, 2, 3}]

Out[7]= 165
```

Ahora calculamos primero la integral entre y=2 e y=3, que llamamos G[x]. Y a continuación calculamos la integral de esta función entre x=1 y x=7. Observamos que se obtiene el mismo resultado.

```
In[8]:= G[x_] = Integrate[f[x, y], {y, 2, 3}]
```

Out[8]= $\frac{39}{2} + 2\,x$

```
In[9]:= Integrate[G[x], {x, 1, 7}]

Out[9]= 165
```

Para calcular una aproximación numérica de una integral doble, podemos usar la orden NIntegrate.

```
In[10]:= g[x_, y_] := Sin[x * Sin[y]]

Integrate[g[x, y], {x, 1, 7}, {y, 2, 3}]

NIntegrate[g[x, y], {x, 1, 7}, {y, 2, 3}]
```

Out[12]= $\int_2^3 (\mathrm{Cos}[\mathrm{Sin}[y]]\,\mathrm{Csc}[y] - \mathrm{Cos}[7\,\mathrm{Sin}[y]]\,\mathrm{Csc}[y])\, dy$

```
Out[14]= 2.19202
```

Ejercicio: Calcule la integral de la función f(x,y)=x^2*y^3+x en el dominio [2,5]×[3,7].

Ejercicio: Calcule la integral de la función g(x,y)=sen(x*cos(y^2)) en el dominio [-2,2]×[-5,7].

■ Representación de regiones regulares. Integrales en dichas regiones.

■ Dominios tipo I.

Los dominios de la forma R={(x, y) $\in \mathbb{R}^{2\,:\, a\le x\le b}$, α (x) ≤ y ≤ β (x)} se dicen dominios regulares de tipo I. La coordenada x varía entre dos números fijos a y b, y, dado un valor de x, la coordenada y de los puntos (x , y), varía entre α(x) y β(x).

Para representar gráficamente uno de estos dominios, dibujaremos la frontera de R. Para lo cual representamos las gráficas de las funciones α(x) y β(x). Como ejemplo consideramos el dominio R={(x, y) $\in \mathbb{R}^{2\,:\, 1\le x\le 2}$, x − 2 ≤ y ≤ −x^2 + 5}.

```
In[15]:= Plot[{x - 2, -x^2 + 5}, {x, 1, 2}]
```




```
Out[15]= - Graphics -
```

Hemos dibujado parte de la frontera del dominio R. Para representar el dominio empleamos la orden FilledPlot que estudiamos en la práctica 7.

In[16]:=
```
Needs["Graphics`FilledPlot`"];

FilledPlot[{x - 2, -x^2 + 5}, {x, 1, 2}]
```



Out[18]= - Graphics -

Ejercicio: Represente la región E1={(x, y) ∈ $\mathbb{R}^{2}$: $1 \le x \le 4$, $-x^2 + 5x - 3 \le y \le x^2 + 7$} .

Supongamos que queremos calcular la integral de la función f(x,y)=x^2+x*y en el dominio R. Para calcular dicha integral debemos considerar los siguientes límites de integración.

$$\int_D f(x, y)\, dx\, dy = \int_{x=1}^{x=2} \left(\int_{y=\alpha(x)}^{y=\beta(x)} f(x, y)\, dy \right) dx$$

Esta integral puede calcularse empleando las siguientes instrucciones de *Mathematica*, la primera emplea la orden Integrate, y la siguiente emplea la paleta BasicInput. Se comprueba que ambas instrucciones dan el mismo resultado.

In[19]:=
```
Integrate[ Integrate[x^2 + x * y, {y, x - 2, -x^2 + 5}], {x, 1, 2}]
```

Out[19]= $\frac{457}{40}$

```
In[20]:= ∫_1^2 ∫_(x-2)^(-x^2+5) (x^2 + x*y) dy dx
```

$$\int_1^2\int_{x-2}^{-x^2+5}(x^2+x*y)\,dy\,dx$$

Out[20]= $\frac{457}{40}$

Ejercicio: Calcule la integral de la función x^2*y+2x*yen la región anterior E1= $\{(x, y) \in \mathbb{R}^{2:\ 1\le x\le 4},\ \ -x^{\wedge}2+5\,x-3 \le y \le x^{\wedge}2+7\}$.

■ Dominios tipo II.

Los dominios de la forma S=$\{(x, y) \in \mathbb{R}^{2:\ c\le y\le d},\ \ \gamma(y) \le x \le \delta(y)\}$ se dicen dominios regulares de tipo II. Para dibujar la frontera de estos dominios debemos dibujar las rectas y=c, y=d, así como las gráficas de las curvas (dadas de forma implícita) x=γ(y), x=δ(y). Estas últimas gráficas se pueden representar con la orden ImplicitPlot.

Como ejemplo representamos el dominio $S = \{(x, y) \in \mathbb{R}^{2:\ -1\le y\le 2},\ \ -(1/2)\,y-1 \le x \le 4-y^{\wedge}2\}$.

Para representar este dominio usando ImplicitPlot, debemos especificar un intervalo para la variable x lo suficientemente grande como para representar completamente el dominio S. Una vez representado S, se puede ajustar el intervalo para x.

```
In[21]:= Needs["Graphics`ImplicitPlot`"]

A1 = ImplicitPlot[x == -(1/2) y - 1, {x, -4, 4}]

B1 = ImplicitPlot[x == 4 - y^2, {x, -4, 4}]

C1 = Plot[{-1, 2}, {x, -4, 4}]

Show[A1, B1, C1]
```



Out[23]= - Graphics -



Out[25]= - Graphics -



Out[27]= - Graphics -



Out[29]= - Graphics -

A la vista de la gráfica, podemos cambiar intervalo de definición de x

In[30]:=
```
A2 = ImplicitPlot[x == -(1/2) y - 1, {x, -2.5, 0}]

B2 = ImplicitPlot[x == 4 - y^2, {x, -3, 4}]

C2 = Plot[{-1, 2}, {x, -2, 4}]

Show[A2, B2, C2]
```



Out[30]= - Graphics -



Out[32]= - Graphics -




```
Out[34]= - Graphics -
```




```
Out[36]= - Graphics -
```

Ejercicio: Represente la región E2= $\left\{(x, y) \in \mathbb{R}^{2} : 2-\sqrt{-8+6y-y^2} \leq x \leq 7+\sqrt{-8+6y-y^2},\ \ 2 \leq y \leq 4\right\}$.

Si queremos calcular la integral de la función g(x,y)=x*cos(y)en el dominio S, debemos considerar los límites de integración que se muestran a continuación.

$$\int_S g\ (x,\ y)\ dx\,dy = \int_{y=-1}^{y=2}\int_{x=-(1/2)\ y-1}^{x=4-y^2} g\ (x,\ y)\ dx\,dy$$

Para efectuar la integral con *Mathematica*, podemos emplear alguna de las instrucciones siguientes.

```
In[37]:= Integrate[ Integrate[x * Cos[y], {x, - (1 / 2) y - 1, 4 - y^2}] , {y, -1, 2}]
```

$$\text{Out[37]=}\ -\frac{71\,\text{Cos}[1]}{4} - 25\,\text{Cos}[2] + \frac{149\,\text{Sin}[1]}{8} - \frac{23\,\text{Sin}[2]}{4}$$

In[38]:= $\int_{-1}^{2}\int_{-(1/2)\,y-1}^{4-y^{\wedge}2} \mathbf{x * Cos[y]}\,\mathrm{d}x\,\mathrm{d}y$

Out[38]= $-\frac{71\,\mathrm{Cos}[1]}{4} - 25\,\mathrm{Cos}[2] + \frac{149\,\mathrm{Sin}[1]}{8} - \frac{23\,\mathrm{Sin}[2]}{4}$

Ejercicio: Calcule la integral de la función h(x,y)=x*y en la región del ejercicio anterior E2= $\left\{(x, y) \in \mathbb{R}^{2}: 2-\sqrt{-8+6\,y-y^2} \leq x \leq 7+\sqrt{-8+6\,y-y^2}\ ,\ \ 2 \leq y \leq 4\right\}$.

■ Dominios tipo III.

Las regiones de tipo III son aquellas que pueden considerarse como de tipo I o como de tipo II. Por ejemplo, consideramos la región T={(x, y) $\in \mathbb{R}^2$:(x-2)^2+(y-2)^2=3^2}que es un círculo de centro el punto (2,2) y radio 3. Esta región se puede representar como una región de tipo I (despejando y en función de x) o de tipo II (despejando x en función de y).

Podemos representar la frontera del dominio (es decir, la circunferencia) usando la orden ImplicitPlot.

In[39]:= **ImplicitPlot[(x - 2)^2 + (y - 2)^2 == 3^2, {x, -5, 5}]**



Out[39]= - Graphics -

En ocasiones es preferible representar la circunferencia con la orden Circle, que dibuja una circunferencia dado su centro y su radio. La opción AspectRatio hace que *Mathematica* elija automáticamente la razón entre la altura y la anchura del dibujo, de forma que represente una circunferencia.

```
In[40]:= A = Graphics[Circle[{2, 2}, 3] ]

         Show[A]

         B = Graphics[Circle[{2, 2}, 3] , AspectRatio -> Automatic ]

         Show[B]

Out[40]= - Graphics -
```



```
Out[42]= - Graphics -

Out[44]= - Graphics -
```



```
Out[46]= - Graphics -
```

■ Dominios compuestos.

También se pueden considerar regiones que se descomponen en un número finito de regiones de los tipos I, II o III vistas anteriormente.

Representamos ahora la región W delimitada superiormente por las curvas C1≡(x-2)^2+y^2=4, C2≡ 4-x=y^2, e inferiormente por la recta x=2 y.

```
In[47]:= C1 = Graphics[Circle[{2, 0}, 2], AspectRatio → Automatic]

         C2 = ImplicitPlot[4 - x == y^2, {x, 0, 4}]

         C3 = ImplicitPlot[2 y == x, {x, 0, 4}]

         Show[C1, C2, C3]

Out[47]= - Graphics -
```




```
Out[49]= - Graphics -
```



Out[51]= - Graphics -



Out[53]= - Graphics -

En el dibujo se observan dos regiones limitadas inferiormente por la recta 2y=x. De estas dos regiones, sólo una está limitada superiormente por las curvas C1 y C2.

Para integrar en muchos dominios, es necesario calcular los puntos de intersección de las curvas que determinan dicho dominio. Por ejemplo, en la región anterior, calculamos con la orden Solve el punto de intersección de C1≡(x-2)^2+y^2=4 con C2≡ 4-x=y^2.

In[54]:= **Solve[{(x - 2)^2 + y^2 == 4, 4 - x == y^2}, {x, y}]**

Out[54]= $\left\{\left\{x \to 1,\ y \to -\sqrt{3}\right\},\ \left\{x \to 1,\ y \to \sqrt{3}\right\},\ \{x \to 4,\ y \to 0\},\ \{x \to 4,\ y \to 0\}\right\}$

De los tres puntos de corte, sólo nos interesa el que está en la frontera de la región considerada, es decir $(x,\ y) = \left(1,\ \sqrt{3}\right)$.

Calculamos de forma similar la intersección de la recta 2y=x con la curva C1≡(x-2)^2+y^2=4.

```
In[55]:= Solve[{(x - 2)^2 + y^2 == 4, 2 y == x}, {x, y}]
```

Out[55]= $\left\{\{y \to 0,\ x \to 0\},\ \left\{y \to \frac{8}{5},\ x \to \frac{16}{5}\right\}\right\}$

De donde obtenemos el punto (0,0).

El tercer punto que vamos a calcular es el punto de corte de la recta 2y=x con la parábola C2≡4-x=y^2.

```
In[56]:= Solve[{2 y == x, 4 - x == y^2}, {x, y}]
```

Out[56]= $\left\{\left\{x \to 2\left(-1-\sqrt{5}\right),\ y \to -1-\sqrt{5}\right\},\ \left\{x \to 2\left(-1+\sqrt{5}\right),\ y \to -1+\sqrt{5}\right\}\right\}$

Y obtenemos el punto de coordenada y positiva, (x,y)=($2\left(-1+\sqrt{5}\right),\ -1+\sqrt{5}$)

Si quisiéramos representar la región empleando FilledPlot, debemos despejar y en función de x, y considerar los puntos de intersección hallados.

In[57]:= W1 = FilledPlot$\left[\left\{x/2,\ \sqrt{4-(x-2)^{\wedge}2}\right\},\ \{x, 0, 1\}\right]$

W2 = FilledPlot$\left[\left\{x/2,\ \sqrt{4-x}\right\},\ \left\{x, 1, 2\left(-1+\sqrt{5}\right)\right\}\right]$

```
Show[W1, W2]
```



Out[57]= - Graphics -




```
Out[59]= - Graphics -
```




```
Out[61]= - Graphics -
```

La región W que hemos representado puede descomponerse como dos regiones de tipo I. Por tanto la integral de la función h(x,y)= 2x*y^2-x en W se puede hallar sumando la integral de h en W1 mas la integral de h en W2.

$$\int_W h(x, y)\, dx\, dy = \int_{W1} h(x, y)\, dx\, dy + \int_{W2} h(x, y)\, dx\, dy$$

Cada una de estas integrales se puede calcular poniendo los límites de integración adecuados, como se vió al estudiar los dominios tipo I. Las siguientes instrucciones calculan la integral en W1 y W2, y las suman para hallar la integral en W.

```
In[62]:= NIntegrate[ Integrate[2 x * y^2 - x, {y, x / 2, √(4 - (x - 2) ^ 2) }] , {x, 0, 1}] +
          NIntegrate[ Integrate[2 x * y^2 - x, {y, x / 2, √(4 - x) }] , {x, 1, 2 (-1 + √5)}]

Out[62]= 3.10028
```

In[63]:= $$N\left[\int_0^1\int_{x/2}^{\sqrt{4-(x-2)^2}} (2\,x*y^2-x)\,dy\,dx + \int_1^{2\,(-1+\sqrt{5})}\int_{x/2}^{\sqrt{4-x}} (2\,x*y^2-x)\,dy\,dx\right]$$

Out[63]= 3.10028

Ejercicio: Calcule la integral de la función t(x,y)=x^4+y^4+2en el dominio W definido en esta sección.

Ejercicio: Represente el dominio acotado EC que se encuentra limitado por las rectas y=x, y=4x-2, y= - x+8.

Ejercicio: Calcule la integral de la función q(x,y)=x+2yen el dominio EC definido en el ejercicio anterior.

■ Representación de regiones mediante coordenadas polares.

Consideremos el dominio R1={(x, y) ∈ $\mathbb{R}^{2:x^{\wedge}2+y^{\wedge}2\leq 1}$, $0 \leq y \leq x$}. Las siguientes instrucciones permiten visualizar esta región.

In[64]:=
```
G1 = FilledPlot[Sqrt[1 - x^2], {x, Cos[Pi / 4], 1}]

G2 = FilledPlot[x, {x, 0, Cos[Pi / 4]}]

Show[G1, G2]
```



Out[64]= - Graphics -



Out[66]= - Graphics -



Out[68]= - Graphics -

Recordamos la función que definía el cambio de coordenadas polares, esto es, F(ρ , φ)=(ρ*cos(φ) , ρ*sen(φ)). La región R1 puede definirse en coordenadas polares como una región T1=={(ρ, φ) $\in \mathbb{R}^{2\,:\,\rho\leq 1}$, $0 \leq \varphi \leq$ Pi / 4}, ya que F(T1)=R1.

Vamos a representar la frontera de R1 usando coordenadas polares. Para ello vemos que la parte curva de la frontera de T1 la forman los puntos (ρ,φ) con ρ=1, y con 0≤φ≤Pi / 4. Así pues para estos puntos se tiene que x= ρ*cos(φ)=cos(φ), y= ρ*sen(φ)=sen(φ).

Dibujamos la parte curva de la frontera con la orden ParametricPlot, donde damos la coordenada x, así como la coordenada y, de los puntos a representar en función de φ.

```
In[69]:= ParametricPlot[{Cos[φ], Sin[φ]}, {φ, 0, Pi / 4}, AxesOrigin → {0, 0},
          AspectRatio → Automatic]
```



```
Out[69]= - Graphics -
```

En este caso el valor de ρ es constantementeigual a 1. En casos más generales la frontera de una región se corresponderá con una curva dada en polares mediante una función $\rho(\varphi)$.

Por ejemplo, supongamos que queremos representar los puntos que tienen radio $\rho(\varphi)$=2*sen(3φ). Es decir, puntos del plano x=2* sen(3φ)*cos(φ), y=2*sen(3φ)*sin(φ). Estos puntos se representan con la siguiente instrucción.

```
In[70]:= ParametricPlot[{2 Sin[3 φ] * Cos[φ], 2 Sin[3 φ] * Sin[φ]}, {φ, 0, 2 * Pi}, AspectRatio → 1]
```




```
Out[70]= - Graphics -
```

En ocasiones es conveniente usar otra notación. Por ejemplo el mismo gráfico anterior se puede representar mediante esta instrucción.

```
In[71]:= r[t_] := 2 Sin[3 t]

         ParametricPlot[{r[t] * Cos[t], r[t] * Sin[t]}, {t, 0, 2 * Pi}, AspectRatio → 1]

         \(\)
```




```
Out[73]= - Graphics -

Out[75]=
```

De igual forma representamosestas otras funciones.

```
In[76]:= ParametricPlot[{2 * (1 + Cos[t]) * Cos[t], 2 * (1 + Cos[t]) * Sin[t]}, {t, 0, 2 * Pi},
          AspectRatio → 1]
```




```
Out[76]= - Graphics -
```

```
In[77]:= ParametricPlot[{√(2*3^2*Cos[2 t]) *Cos[t], √(2*3^2*Cos[2 t]) *Sin[t]},
          {t, 0, Pi/4}, AspectRatio → 1]
```




```
Out[77]= - Graphics -
```

Ejercicio: Represente la curva dada por la función en polares r(t)=3sin(t)*cos(t),para 0≤t≤2 Pi. Idem r(t)=4cos(t)sen^2(t) para 0≤t≤Pi